
Przyk lady prostych zdań i ich reprezentacja w Prologu

Zdania wyrażaja֒ce fakty można zapisać jako klauzule Prologu. Zwróćmy na
pocza֒tek uwage֒, by wszystkie klauzule by ly pisane z ma lej litery. Praktyczne jest
również upraszczanie polskiej fleksji:

Mam kolege Zdzicha.

Mam kolege Rycha.

Zdzich ma ksiazke.

Zdzich ma Mercedesa.

Zdzich ma Mariole.

Zdzich lubi Mercedesa.

Rychu lubi wino.

Zona Rycha lubi wino.

kolega_zdzich.

kolega_rychu.

zdzich_ma_ksiazka.

zdzich_ma_mercedes.

zdzich_ma_mariola.

zdzich_lubi_mercedes.

rychu_lubi_wino.

zona_rycha_lubi_wino.

Prolog może zapamie֒tać te fakty, a naste֒pnie pytany o nie odpowie twierdza֒co,
a pytany o jakiekolwiek inne fakty odpowie przecza֒co.

?- kolega_zdzich.

yes

?- kolega_roman.

no

Jednak ten sposób wyrażania faktów posiada szereg wad. Na przyk lad,
gdybyśmy zapytali o ostatni fakt na líscie w formie: rycha_zona_lubi_wino
to Prolog nie zauważy lby zwia֒zku tego faktu z wcześniej wprowadzonym
zona_rycha_lubi_wino:

?- zona_rycha_lubi_wino.

yes

?- rycha_zona_lubi_wino.

no

Dlatego do wyrażania faktów lepiej użyć struktur:

kolega(zdzich).

kolega(rychu).

ma(zdzich, ksiazka).

ma(zdzich, mercedes).

ma(zdzich, mariola).

lubi(zdzich, mercedes).

lubi(rychu, wino).

lubi(zona(rychu), wino).

?- kolega(zdzich).

yes

?- lubi(X, wino).

X = rychu ? ;

X = zona(rychu)

yes

Struktury i predykaty

Prolog pozwala na wprowadzanie faktów strukturalnych, co daje duża֒ swobode֒.
W korzystaniu z tej swobody, podobnie jak z innych swobód, warto stosować
jednak pewna֒ dyscypline֒, np.:

Czy zamiast pisać tak:

kolega(zdzich).

kolega(rychu).

ma(zdzich, ksiazka).

ma(zdzich, mercedes).

ma(zdzich, mariola).

lubi(zdzich, mercedes).

lubi(rychu, wino).

lubi(zona(rychu), wino).

nie moglibyśmy równie dobrze tak:

kolega(zdzich).

rychu(kolega).

zdzich(ma, ksiazka).

mercedes(ma, zdzich).

ma(mariola, zdzich).

zona(rychu, lubi, wino).

zona(rychu, lubi(wino)).

lubi(zona(rychu(wino))).

lubi(zona, rychu, wino).

Odpowiedź: formalnie owszem, praktycznie nie bardzo.

Struktura najwyższego poziomu (zewne֒trzna) traktowana jest przez Prolog jako
symbol relacji wyrażaja֒cej jakís zwia֒zek mie֒dzy jej argumentami.

Struktury wewne֒trzne (moga֒ być dowolnie zagnieżdżone) traktowane sa֒ jako
funkcje określaja֒ce obiekt pozostaja֒cy w zależności od innych obiektów.

Zapis relacji wyraża fakt logiczny, i nazywany jest predykatem.
Przekszta lcaja֒c fakty wyrażone w normalnym je֒zyku na predykaty logiczne
najcze֒ściej używamy orzeczenia zdania (czasownika) jako symbolu predykatu.
Podmiot staje sie֒ argumentem predykatu, podobnie jak dope lnienia.

Zatem predykat powinien mieć sta la֒ liczbe֒ argumentów, które powinny mieć
swoje określone role, aczkolwiek moga֒ istnieć podobne predykaty z różna֒ liczba֒
argumentów, np.:

ma2( kto, co )

ma3( kto, co, kiedy )

ma4( kto, co, kiedy, gdzie )

W Prologu takie predykaty moga֒ mieć jednakowa֒ nazwe֒, ponieważ Prolog
potrafi je odróżnić po liczbie argumentów: ma/2, ma/3, ma/4

Zmienne w Prologu

Symbol termu rozpoczynaja֒cy sie֒ wielka֒ litera֒ (lub podkreślnikiem ) jest
zawsze zmienna֒ w Prologu. Symbol predykatu nie może być zmienna֒ i nie może
zaczynać sie֒ wielka֒ litera֒. Zmienna w aksjomacie jest traktowana jako
kwantyfikowana uniwersalnie, a zmienna w zapytaniu jako kwantyfikowana
egzystencjalnie. Zakresem zmiennych jest ca la klauzula w której wyste֒puja֒.

Zawartość bazy danych:

/* some like wine */

likes(ed, wine).

likes(wife(ed), wine).

/* everyone likes beer */

likes(X, beer).

/* dick likes his merc */

likes(dick, mercedes).

Odpowiadanie na pytania:
?- likes(X, wine).

X = ed ? ;

X = wife(ed)

yes

?- likes(ed, X).

X = wine ? ;

X = beer

yes

?- likes(X, beer), likes(X, mercedes).

X = dick ? ;

no

W pierwszych dwóch zapytaniach zmienna X jest za każdym razem inna֒
zmienna֒. Natomiast w trzecim zapytaniu (o piwie i mercedesie), sa֒ dwa
wysta֒pienia tej samej zmiennej X, wie֒c musza֒ mieć te֒ sama֒ wartość.

Operator unifikacji

Operator unifikacji = porównuje operandy. Jeśli obie sa֒ termami sta lymi, zwraca
logiczny wynik porównania (identyczne czy nie). Jeśli jeden lub oba operandy sa֒
zmiennymi, wtedy wynik jest zawsze prawda֒, z efektem ubocznym przypisania
zmiennej sta lego operandu. Gdy oba operandy by ly zmiennymi to pozostaja֒
nimi, ale sa֒ zunifikowane, czyli musza֒ mieć równa֒ wartość w przysz lości.

?- wife(ed) = meg.

no

?- wife(ed) = X.

X = wife(ed)

yes

?- wife(X) = wife(Y).

Y = X

yes

/* no artihmetic */

?- 2 + 3 = 5.

no

?- 2 + 3 = X.

X = 2+3

yes

/* dont try */

?- father(son(X)) = X.

Jak widać w przyk ladach po lewej stronie, unifikacja jest elastyczna i porównuje
wyrażenia strukturalnie, przypisuja֒c zmienne by dopasować wyrażenia. Jednak
żadne obliczenia arytmetyczne nie sa֒ wykonywane.

Wprowadzanie faktów i odpowiadanie na pytania

Zapisane w postaci klauzul fakty można wprowadzić do Prologu, który
przyjmuje je jako aksjomaty, umieszcza (po kolei) w swojej bazie danych,
i zaczyna w nie wierzyć (bezgranicznie).

Można również wprowadzać fakty (w tym samym formacie, zakończone kropka֒)
jako zapytania, na które Prolog ma odpowiedzieć. W czasie normalnej pracy
Prolog jest w laśnie w takim trybie odpowiadania na pytania. Aby wprowadzić
aksjomaty używamy specjalnego predykatu consult/1, który wczytuje fakty
z podanego pliku, albo z wej́scia: consult(user). Pojedyncze fakty można
również wprowadzać predykatami asserta i assertz (patrz dalej).

Prolog odpowiada na pytania przez przeszukiwanie swojej bazy danych,
w kolejności wprowadzonych aksjomatów, dopasowuja֒c predykat i kolejne
argumenty zapytania do argumentów aksjomatów.

Pos lugiwanie sie֒ gprolog-iem

gprolog jest  latwo doste֒pnym interpreterem Prologu. Można go wywo lać
w taki sposób, żeby od razu na starcie wczyta l fakty zawarte w określonym pliku:

> gprolog --init-goal "consult(’zdzich2.pro’)"

compiling zdzich2.pro for byte code...

zdzich2.pro compiled, 10 lines read - 1004 bytes written, 15 ms

GNU Prolog 1.2.18

By Daniel Diaz

Copyright (C) 1999-2004 Daniel Diaz

| ?- kolega(zdzich).

(1 ms) yes

gprolog posiada szereg rozszerzeń w odniesieniu do standardu Prologu, jak
również szereg zmiennych konfiguracyjnych. Na przyk lad, w przypadku
zapytania o fakt, którego symbol predykatu nie jest znany, gprolog domyślnie
generuje b la֒d.

Można ustawić flage֒ konfiguracyjna֒, aby takie zapytanie otrzymywa lo po prostu
odpowiedź negatywna֒:

| ?- jest_fajnie.

uncaught exception: error(existence_error(procedure,jest_fajnie/0),top_level/0)

| ?- set_prolog_flag(unknown, fail).

yes

| ?- jest_fajnie.

no

Ponieważ set_prolog_flag jest predykatem (jak wszystko w Prologu), wie֒c
wywo luje sie֒ go w trybie zadawania pytań. Próba wprowadzenia go z pliku
w trybie consult by laby równoważna próbie przedefiniowania predykatu
wbudowanego, co jest niedopuszczalne.

SWI Prolog

Podobnie można wywo lać inny wygodny interpreter Prologu — SWI Prolog:

> pl -f zdzich2.pro

% /home/witold/cla/ai/Prolog/zdzich2.pro compiled 0.00 sec, 2,768 bytes

Welcome to SWI-Prolog (Version 5.6.6)

Copyright (c) 1990-2005 University of Amsterdam.

SWI-Prolog comes with ABSOLUTELY NO WARRANTY. This is free software,

and you are welcome to redistribute it under certain conditions.

Please visit http://www.swi-prolog.org for details.

For help, use ?- help(Topic). or ?- apropos(Word).

1 ?- kolega(X).

X = zdzich ;

X = rychu ;

No

Ćwiczenie

Uruchom jakís system Prologu, wprowadź kilka prostych faktów (z argumentami
oraz bez) za pomoca֒ consult(user). Zadaja֒c pytania sprawdź poprawność
wprowadzonych danych. Za pomoca֒ klawisza ; (średnika) wymuś wyświetlanie
wszystkich możliwych wartości jakiej́s zmiennej.

Pod has lem "prolog tutorial" wygoogluj kilka stron z wprowadzeniem do
Prologu. Skopiuj znalezione tam proste przyk ladowe programy, np. hanoi,
i spróbuj je uruchomić zgodnie z za la֒czonymi wyjaśnieniami.

Na przyk lad:
http://www.learnprolognow.org/

http://www.cs.bham.ac.uk/~pjh/prolog_course/se207.html

Regu ly

Oprócz faktów prostych, bezwarunkowych, możemy wyrażać w Prologu fakty
zawieraja֒ce spójnik implikacji (wstecznej):

lubi(zdzich, X) :- lubi(X, wino).

Fakty tego typu nazywa sie֒ regu lami.

Prolog poszukuje odpowiedzi na zadawane pytania dopasowuja֒c pytanie do
kolejnych faktów w bazie danych. Gdy pytanie dopasowuje sie֒ do faktu prostego,
odpowiedź jest natychmiastowa. Gdy fakt jest regu la֒ to Prolog próbuje
dopasować pytanie do jej lewej strony. Gdyby to sie֒ uda lo, Prolog wywo luje sie֒
rekurencyjnie, aby udowodnić fakty po prawej stronie regu ly. Sukces wywo lania
rekurencyjnego oznacza odpowiedź pozytywna֒ na oryginalne pytanie.

likes(dick, X) :- likes(X, wine).

likes(dick, mercedes).

likes(dick, wine).

likes(ed, wine).

likes(wife(ed), wine).

?- likes(dick,X).

X = dick ;

X = ed ;

X = wife(ed) ;

X = mercedes ;

X = wine.

Spójniki i konotacje logiczne

Operator :- wyste֒puja֒cy w regu lach można traktować jako spójnik, ponieważ
pozwala on tworzyć klauzule z lożone z prostych. W sensie logicznym odpowiada
on implikacji skierowanej wstecz ⇐. Implikacje֒ można stosować tylko
w definicjach regu l (nie w zapytaniach), i to tylko raz.

W Prologu istnieja֒ jeszcze spójniki koniunkcji ∧, zapisywanej przecinkiem, oraz
alternatywy logicznej ∨, zapisywanej średnikiem:

?- ma(zdzich, mercedes), ma(zdzich, alfa_romeo_8c).

No

?- ma(zdzich, mercedes); ma(zdzich, alfa_romeo_8c).

true

Koniunkcje֒ i alternatywe֒ można stosować tylko po prawej stronie regu l
(w poprzednikach implikacji). Lewa strona regu ly musi być termem atomowym.
Taki formu ly logiczne, bez alternatyw, albo z alternatywami wy la֒cznie
negatywnych litera lów, albo z najwyżej jednym litera lem pozytywnym, nazywa
sie֒ klauzulami Horna.

Zatem można stwierdzić, że Prolog operuje jedynie klauzulami Horna.

Wyjaśnienie proceduralne

Wyrażanie faktów po la֒czonych koniunkcja֒ jest równoważne zapisywaniu tych
faktów oddzielnie (baza danych Prologu jest jedna֒ wielka֒ koniunkcja֒ logiczna֒),
a wie֒c nie jest de facto potrzebne. Jednak wyrażanie faktów po la֒czonych
alternatywa֒ daje możliwości, których nie da sie֒ bez alternatywy uzyskać:

ma(zdzich, pieniadze); ma(zdzich, alfa_romeo_8c).

Ten fakt wyraża ważna֒ w lasność pienie֒dzy: albo sie֒ je trzyma, albo sie֒ je
wydaje. Ale jak Prolog mia lby korzystać z takiego faktu, w wyszukiwaniu
odpowiedzi na zapytanie? Nie może ani skorzystać z cze֒ści pierwszej,
i odpowiedzieć, że Zdzich ma gotówke֒, bo nie wie tego na pewno, ani z cze֒ści
drugiej, z tego samego powodu.

Takiego problemu nie ma, gdy alternatywa wyste֒puje po prawej stronie regu ly:

jest_cool(X) :- ma(X, pieniadze); ma(X, alfa_romeo_8c).

Aby udowodnić, że ktoś jest
”
cool” (ktokolwiek, niekoniecznie nasz bohater

Zdzich), wystarczy sprawdzić, że ma kase֒, lub alfe֒.

Czyli można zapisywać te klauzule, z którymi radzi sobie algorytm wyszukiwania.

Konwersja faktów logicznych do Prologu

Rozważmy proces zamiany faktów wyrażonych w je֒zyku predykatów pierwszego
rze֒du na klauzule Prologu. W ogólności należy pamie֒tać, że sta le, funkcje, oraz
symbole predykatów musza֒ zaczynać sie֒ od ma lej litery, a zmienne od wielkiej.

Negacja, a może raczej jej brak

W Prologu nie ma spójnika negacji. Jednak istnieje wbudowany predykat not,
którego znaczenie można określić jako:

”
nie da sie֒ udowodnić, że ...”.

W niektórych przypadkach można go używać w charakterze negacji, ale niekiedy
daje on nieoczekiwane wyniki.

Zawartość bazy danych:

man(dick).

dog(spot).

?- man(X).

X = dick ?

yes

?- not(man(spot)).

yes

?- not(man(X)).

no

Można by loby oczekiwać, że Prolog znajdzie indywiduum, które nie jest
cz lowiekiem.

Można by loby oczekiwać, że skoro da lo sie֒ udowodnić not(man(spot)) to tym
bardziej powinno dać sie֒ udowodnić not(man(X)).

Jedno i drugie oczekiwanie zawodzi. Można jedynie wycia֒gna֒ć wniosek, że not

jest dziwna֒, nieintuicyjna֒ forma֒ przeczenia.

Predykatu negacji not można używać tylko w zapytaniach, a nie
w stwierdzeniach zapamie֒tywanych w bazie danych.

Wyjaśnienie dlaczego tylko taka forma przeczenia jest doste֒pna w Prologu
pojawi sie֒ później, a na razie musimy przyja֒ć, że obowia֒zuje myślenie
pozytywne, i staramy sie֒ nic nie negować.

Jednak zwróćmy uwage֒, że sam Prolog jest na ladowany myśleniem
negatywnym, bo zaprzecza wszystkiemu, co nie jest dla niego oczywiste po
sprawdzeniu swojej bazy danych. Te֒ w lasność, negowania wszystkiego co nie
jest jawnie znane, nazywa sie֒ za lożeniem świata zamknie֒tego
(Closed-World Assumption, CWA).

W wielu interpreterach Prologu predykat not nie wyste֒puje jako taki. Zamiast
niego jest doste֒pny operator \+ o takim samym dzia laniu.

Obliczenia na strukturach

To co nazywamy w Prologu struktura֒, czyli zapis predykatu z argumentami,
można traktować jako strukture֒ danych, i budować z ich użyciem obliczenia.

Rozważamy naste֒puja֒ca֒ arytmetyke֒, gdzie wprowadzamy liczby za pomoca֒
symbolu zero i struktury s(X), która oznacza naste֒pnik (naste֒pna֒ liczbe֒ po)
X. Jest to tzw. arytmetyka Peano. Na przyk lad, liczbe֒ 5 zapisujemy w postaci:
s(s(s(s(s(zero))))). Chcemy zdefiniować dodawanie za pomoca֒ predykatu
dodaj(Skl1, Skl2, Suma) prawdziwego zawsze, gdy Suma jest suma֒
podanych dwóch sk ladników, i fa lszywego w pozosta lych przypadkach:

suma(zero, Skl2, Skl2).

suma(s(X), Skl2, s(Suma)) :- suma(X, Skl2, Suma).

Aby prowadzić obliczenia w tej arytmetyce musimy pos lugiwać sie֒ notacja֒
naste֒pników, np. żeby obliczyć 3+4:

?- suma(s(s(s(zero))), s(s(s(s(zero)))), X).

X = s(s(s(s(s(s(s(zero)))))))

Zdefiniowanie mnożenia wymaga troche֒ wie֒cej zachodu, spróbuj!

Obliczenia
”
wstecz”

Zauważmy, że w zdefiniowanym predykacie suma dwa pierwsze argumenty
stanowia֒ dane, a trzeci argument stanowi wynik obliczeń. Jest tak, podobnie jak
w innych je֒zykach programowania, w których funkcje moga֒ mieć argumenty
typu

”
out” i zwracać w nich wyniki. Podobnie, jak w tych innych je֒zykach

”
funkcja”nie jest dok ladnie funkcja֒ w sensie matematycznym, tak w Prologu

predykat nie jest dok ladnie predykatem w sensie logicznym.

Jednak Prolog nie ma mechanizmu deklarowania, który argument jest typu

”
out”, zatem co by sie֒ sta lo, gdybyśmy zamiast zadawać proste pytania typu

3+4=? zacze֒li zadawać równania do rozwia֒zania, typu 3+?=4:

?- suma(s(s(s(zero))), X, s(s(s(s(zero))))).

X = s(zero) ;

No

Dobrze, jedynym rozwia֒zaniem powyższego równania jest s(zero), i nic
innego. Ta zdolność do prowadzenia obliczeń

”
wstecz” jest efektem ubocznym

prologowego algorytmu przeszukiwania bazy danych i dopasowywania wzorców.

Oczywíscie możemy posuna֒ć sie֒ dalej, i zadać pytanie, na które odpowiedź nie
jest jednoznaczna: ?+?=4. Uzyskamy wszystkie rozk lady liczby 4 na sk ladniki:

?- suma(X, Y, s(s(s(s(zero))))).

X = zero

Y = s(s(s(s(zero)))) ;

X = s(zero)

Y = s(s(s(zero))) ;

X = s(s(zero))

Y = s(s(zero)) ;

X = s(s(s(zero)))

Y = s(zero) ;

X = s(s(s(s(zero))))

Y = zero ;

No

Maja֒c zdefiniowane mnożenie moglibyśmy dokonywać faktoryzacji liczb, a nawet
wycia֒gać pierwiastki! Spróbuj.

Prawdziwe liczby

Prolog potrafi pos lugiwać sie֒ prawdziwymi liczbami, porównywać je, i obliczać
wyrażenia liczbowe, choć to ostatnie robi nieche֒tnie. Sprawdźmy to,
wykorzystuja֒c operator porównania =.

?- 0 = 0.

Yes

?- 0 = 1.

No

?- 2+2 = 4.

No

?- 2+2 = X.

X = 2+2

Yes

Prolog uważa, że jego podstawowym zadaniem jest przeszukiwanie bazy danych
i dopasowywanie termów, i nie be֒dzie zawraca l sobie g lowy obliczaniem
wartości, gdy któryś z termów jest wyrażeniem liczbowym. Zakle֒ciem, które
zmusza Prolog do wykonania obliczeń jest operator is, który oblicza wyrażenie
po prawej stronie i podstawia (lub porównuje) pod zmienna֒ po lewej stronie:

rowna_sie2(X, Y) :- X1 is X, X1 = Y.

Mamy teraz wyniki dobre:

?- rowna_sie(2+2, 4).

Yes

?- rowna_sie(2+2, X).

X = 4

ale również nadal nieakceptowalne, albo b le֒dy:

?- rowna_sie(2+2,2+2).

No

?- rowna_sie(X, 2+2).

ERROR: (user://4:136):

is/2: Arguments are not sufficiently instantiated

Porażka w pierwszym przyk ladzie po prawej wynika z nieobliczenia drugiego
argumentu. Jednak jak pokazuje ostatni przyk lad, obliczenia prowadzone przez
is nie toleruja֒ niepodstawionego argumentu po prawej stronie, zatem is musi
być stosowane ostrożnie, po zbadaniu postaci posiadanych argumentów: czy sa֒
wartościami, czy wyrażeniami, czy podstawione, czy nie. Prolog posiada szereg
mechanizmów do sprawdzania tej postaci (patrz dalej).

Operatory infiksowe

Zasadniczo Prolog stosuje zapis formu l (zwanych strukturami) w notacji
funkcyjnej, czyli symbol operacji i lista argumentów w nawiasach okra֒g lych,
oddzielonych przecinkami. Jest jednak dopuszczalne użycie sk ladni
operatorowej, czyli argumenty rozdzielone, poprzedzone, lub poprzedzaja֒ce
symbol operatora, bez nawiasów ani przecinków. Prolog dopuszcza zapis:

a + b jako równoważna֒ alternatywe֒ zapisu: +(a,b)

Wyrażenia te sa֒ ca lkowicie równoważne, a wre֒cz identyczne, ponieważ to
pierwsze traktowane jest jako pewna dodatkowa forma zapisu, i konwertowane
do postaci po prawej w czasie parsowania przez Prolog.

Operatory tego typu jak + można również definiować w programach, co pozwala
na pos lugiwanie sie֒ dowolnymi symbolami operatorów. Na przyk lad możemy
wprowadzić symbol prefiksowego unarnego minusa (lub przeczenia) oraz symbol
infiksowego operatora pote֒gowania:

?-op( 9, fx, ~ ). /* operator minus (unarny) */

?-op(10, yfx, ^ ). /* operator potegowania */

Wtedy każde wyrażenie postaci a ^ b be֒dzie przez Prolog konwertowane do
postaci ^(a,b) i obliczane zgodnie z istnieja֒cymi definicjami predykatu ^

Operatory porównania w Prologu

Równość albo równoważność posiada wiele oblicz w Prologu. Poza operatorem
unifikacji =, który wykonuje porównanie strukturalne z unifikacja֒ zmiennych,
istnieja֒ porównania numeryczne, które pozwalaja֒ obliczać wartości wyrażeń
arytmetycznych. Wymagaja֒ one by obliczane numerycznie termy by ly w pe lni
podstawione, i mia ly wartość liczbowa֒:

X is Y — operand prawostronny Y może być wyrażeniem arytmetycznym,
którego wartość liczbowa jest dopasowana do operandu lewostronnego X,
który może być zmienna֒

X =:= Y — wartości arytmetyczne wyrażeń X i Y sa֒ równe

X =\= Y — wartości arytmetyczne wyrażeń X i Y sa֒ różne

Ponadto, istnieja֒ porównania strukturalne, które nie wyliczaja֒ wartości
liczbowej, a wymagaja֒ pe lnej, literalnej identyczności:

X == Y — termy X i Y sa֒ identyczne, maja֒ identyczna֒ strukture֒ i identyczne
argumenty z dok ladnościa֒ do nazwy, np. X==Y jest zawsze nieprawda֒

X \== Y — termy X i Y nie sa֒ identyczne

Prolog — obliczenia liczbowe 29



?- 3+4 = 4+3.

no % structures differ

?- 3+4 = 3+4.

yes

?- X = 3+4.

X = 3+4

yes

?- 3+X = 3+4.

X = 4

yes

?- 3+4 == 4+3.

no

?- 3+X == 3+4.

no

?- +(3,X) == 3+X.

yes

?- 3+4 \== 4+3.

yes

?- X is 3+4.

X = 7

yes

?- X = 7, X is 3+4.

X = 7

yes

?- X is 3+4, X = 7.

X = 7

yes

?- 3+4 is 4+3.

no % left arg.must be unassigned var.

% or evaluate to a number

?- 3+4 =:= 4+3.

yes % calculates both values

?- X =:= 3+4.

error % both args must have values

?- a =:= 3+4.

error % and they must be arithm.values

?- 3+4 =\= 4+3.

no

Listy

Prolog ma jedna֒ prawdziwa֒ strukture֒ danych jaka֒ jest lista. Lista jest
sekwencja֒ elementów, które moga֒ być atomami, ba֒dź listami. Listy zapisujemy
w nawiasach kwadratowych oddzielaja֒c poszczególne elementy przecinkami, np.:

[a]

[X,Y]

[1,2,3,4]

[a,[1,X],[],[],a,[a]] /* ta lista ma 6 elementow */

Listy można również zapisywać podaja֒c
”
g lowe֒”(element pocza֒tkowy) i

”
reszte֒”

listy, co ma duże znaczenie gdy ta reszta zapisana jest za pomoca֒ zmiennej, np.:

[a|R] /* ta lista ma co najmniej 1 element,R moze byc []*/

[1|[2|[3|[4|[]]]]] /* dokladnie rowna liscie [1,2,3,4] */

[1|[2|[3|[4]]]] /* inny sposob zapisu listy [1,2,3,4] */

[1,2|[3,4]] /* inny dopuszczalny zapis listy [1,2,3,4] */

Liste֒ w notacji [G lowa|Reszta] można również zapisać jako strukture֒ .(G lowa,Reszta)

(nazwa֒ termu jest kropka).

Pomimo iż wie֒kszość wspó lczesnych interpreterów Prologu posiada wiele
operacji na listach (przyk ladowe predykaty zdefiniowane poniżej nazywaja֒ sie֒
odpowiednio: member i append), jest pouczaja֒ce przestudiowanie
rekurencyjnych implementacji podstawowych takich operacji.

Predykat element sprawdza, czy coś jest elementem listy:

element(X,[X|_]).

element(X,[_|Y]) :- element(X,Y).

Ten predykat z la֒cza dwie listy i unifikuje z trzecim argumentem.

merge([],X,X).

merge([X|Y],Z,[X|Q]) :-

merge(Y,Z,Q).

Wypróbuj poniższe zapytania:
?- merge([a,b,c],[w,x,y,z],L).

?- merge([a,b],Y,[a,b,c,d]).

?- merge(Y,[c,d],[a,b,c,d]).

?- merge([b,c],Y,[a,b,c,d]).

?- merge(X,Y,[a,b,c,d]).

?- merge(X,Y,Z).

Spróbuj: napisz definicje֒ predykatu określaja֒cego ostatni element listy.

Debugowanie programów

Prolog zawiera kilka predykatów wspomagaja֒cych analize֒ programów
i umożliwiaja֒cych śledzenie ich wykonania:

spy/1 — ustawia śledzenie wykonania danego predykatu, który można podać
w formie: pred/n wyróżniaja֒c wersje֒ o danej liczbie argumentów,

trace/0 — w la֒cza śledzenie wszystkiego,

nospy/1, notrace/0 — kasuje śledzenie,

nodebug/0 — kasuje wszystkie spy,

debugging/0 — listuje wszystkie spy,

listing/1 — wyświetla komplet definicji dla jednego konkretnego predykatu,

listing/0 — wyświetla wszystkie definicje posiadanych predykatów (poza
predykatami wbudowanymi, które nie maja֒ definicji źród lowej).

Przyk lad

Można zdefiniować
”
zdanie” je֒zyka polskiego jako liste֒ s lów w szyku pasuja֒cym

do gramatyki naszego pie֒knego je֒zyka, w uproszczonej wersji:

rzeczownik(adam).

rzeczownik(stolarz).

rzeczownik(murarz).

czasownik(muruje).

czasownik(hebluje).

podmiot(X) :- rzeczownik(X).

orzeczenie(X) :- czasownik(X).

zdanie([X,Y]) :- podmiot(X), orzeczenie(Y).

Ten schemat pozwala sprawdzać przyk lady różnych konstrukcji, czy sa֒ zdaniami
(wy la֒cznie z punktu widzenia gramatyki, nie wnikaja֒c w ich sens):

?- zdanie([stolarz, muruje]).

Yes

?- zdanie([hebluje, stolarz]).

No

Można oczywíscie prosić o uzupe lnienie cze֒ściowo podstawionego zdania, lub
generować ca le zdania:

?- zdanie(X).

X = [adam, muruje] ;

X = [adam, hebluje] ;

X = [stolarz, muruje] ;

X = [stolarz, hebluje] ;

X = [murarz, muruje] ;

X = [murarz, hebluje] ;

Jako ćwiczenie spróbuj rozwina֒ć powyższy schemat zdania tak, aby dopuszcza l
również zdania bardziej skomplikowane, z różnymi okolicznikami (miejsca,
czasu), i/lub zdania z lożone.

Uzupe lnij zasób s lów i wypróbuj swój programik na różnych mniej lub bardziej
rzeczywistych konstrukcjach zdaniowych.

Ćwiczenie — permutacje

W ćwiczeniu z gramatyka֒, listy potrzebne by ly jedynie do tego, by zdania mog ly
być różnej d lugości. W wielu programach konieczne jest jednak analizowanie
zawartości list i skuteczne nimi manipulowanie. Dobrym ćwiczeniem jest
napisanie predykatu permutacja(X,Y), który sprawdza, czy jego argumenty sa֒
listami, z których jedna jest permutacja֒ drugiej, czyli lista֒ sk ladaja֒ca֒ sie֒ z tych
samych elementów (w tych samych ilościach), tylko być może w innej kolejności:

?- permutacja([a,b,c],[b,c,a]).

Yes

?- permutacja([a,a,c],[c,c,a]).

No

Spróbuj napisać taki predykat. Naste֒pnie sprawdź możliwość generacji
wszystkich permutacji jakiej́s listy przez uruchamianie predykatu z jednym
argumentem niepodstawionym (zmienna֒).

?- permutacja([a,b,c],X).

Schemat: Generate and Test

Wiele programów w Prologu można sensownie napisać wed lug pewnego
przydatnego schematu. Rozważmy na pocza֒tek prosty generator, generuja֒cy
kolejne liczby naturalne, jeśli tylko jakís predykat be֒dzie o te liczby w kó lko
prosi l:

liczba(0).

liczba(N) :- liczba(M), N is M + 1.

Aby zademonstrować jego dzia lanie, czyli spowodować wielokrotne wznawianie
obliczeń tego predykatu, możemy wykorzystać zeroargumentowy predykat fail,
który po prostu zwraca fa lsz:

liczba(N), write(N), nl, fail.

Możemy teraz budować programy sk ladaja֒ce sie֒ z generatora potencjalnych
rozwia֒zań jakiegoś abstrakcyjnego zadania, i predykatu testuja֒cego, który
jedynie sprawdza czy proponowany obiekt jest akceptowalnym rozwia֒zaniem:

generate(X), test(X), gotowe(X).

Na przyk lad, w celu generacji liczb pierwszych wystarczy generować po kolei
liczby ca lkowite, i użyć predykatu sprawdzaja֒cego podzielność, a raczej jej brak:

% The sieve of Eratosthenes, from Clocksin & Mellish 2ed p.170

% finding the prime numbers up to 98.

main :- primes(98, X), write(X), nl.

primes(Limit, Ps) :- integers(2, Limit, Is), sift(Is, Ps).

/* integers(F,T,L) puts the integers from F to T into list L */

integers(Low, High, [Low | Rest]) :-

Low =< High, !, M is Low+1, integers(M, High, Rest).

integers(_,_,[]).

/* sift(L1,L2) sifts non-prime numbers from L1, puts rest into L2 */

sift([],[]).

sift([I | Is], [I | Ps]) :- remove(I,Is,New), sift(New, Ps).

/* remove(N,L1,L2) removes from L1 multiples of number N into L2 */

remove(P,[],[]).

remove(P,[I | Is], Nis) :- 0 is I mod P, !, remove(P,Is,Nis).

remove(P,[I | Is], [I | Nis]) :- not(0 is I mod P),!,remove(P,Is,Nis).

Zawieszanie i wznawianie obliczeń

http://www.inf.ed.ac.uk/teaching/courses/aipp/

http://www.inf.ed.ac.uk/teaching/courses/aipp/lecture_slides/

07_Cut.pdf

Odcie֒cie

Odcie֒cie, zapisywane znakiem wykrzyknika (!), jest operatorem maja֒cym
wartość logiczna֒ prawdy, ale jednocześnie blokuja֒cym mechanizm nawracania
Prologu do dalszych punktów wyboru. Przyjrzyjmy sie֒ na przyk ladach, co to
dok ladnie oznacza:

Dla naste֒puja֒cych definicji:

fakt(a).

fakt(b) :- !.

fakt(c).

Każdy z faktów a,b,c, jest indywidualnie spe lniony,
jednak gdy Prolog próbuje wszystkich definicji po
kolei, i po drodze napotka operator odcie֒cia, to nie
może już kontynuować obliczeń i zwraca odpowiedź
negatywna֒.

?- fakt(a).

Yes

?- fakt(b).

Yes

?- fakt(c).

Yes

?- fakt(X).

X = a ;

X = b ;

No

Jak widać poniżej, obecność
odcie֒cia w definicji fakt powoduje
zak lócenie jego wywo lywania przez
inne fakty.

fakt2(X,Y) :- fakt(X), X = Y.

/***************************/

?- fakt2(X,a).

X = a

Yes

?- fakt2(X,b).

X = b

Yes

?- fakt2(X,c).

No

Zmiana kolejności sprawdzanych
warunków opóźnia wykonanie odcie֒cia,
co troche֒ pomaga, ale nadal odcina
ono pewne rozwia֒zania.

fakt3(X,Y) :- X = Y, fakt(X).

/***************************/

?- fakt3(X,c).

X = c

Yes

?- fakt3(X,Y).

X = a

Y = a ;

X = b

Y = b ;

No

To po co jest nam w laściwie potrzebny operator odcie֒cia?

Odcie֒cie — przypadek 1: utwierdza wybór regu ly

Wyobraźmy sobie predykat sum_to s luża֒cy do obliczania sumy liczb od 1 do
jakiej́s wartości. Drugi argument przeznaczony jest na wynik obliczeń.

sum_to( 1, 1 ).

sum_to( N, R ) :-

N1 is N - 1,

sum_to( N1, R1 ),

R is R1 + N.

To rozwia֒zanie dzia la poprawnie, z wyja֒tkiem kilku przypadków specjalnych, na
przyk lad, kiedy użytkownik wywo la program ze z lymi danymi, albo kiedy
nacísnie

”
;”zmuszaja֒c program do wznawiania obliczeń:

?- sum_to(5,X).

X = 15 ;

ERROR: (user://1:22):

Out of local stack

?- sum_to(5,14).

ERROR: (user://1:27):

Out of local stack

To jest przyk lad programu, który nie powinien w ogóle po obliczeniu jednego
wyniku wznawiać obliczeń, ponieważ nie ma innej sumy liczb niż ta pierwotnie
wyliczona. Można to uwzgle֒dnić za pomoca֒ odcie֒cia (wersja po lewej):

sum_to( 1, 1 ) :- !.

sum_to( N, R ) :-

N1 is N - 1,

sum_to( N1, R1 ),

R is R1 + N.

sum_to( N, 1 ) :- N < 1, !, fail.

sum_to( 1, 1 ).

sum_to( N, R ) :-

N1 is N - 1,

sum_to( N1, R1 ),

R is R1 + N.

Pozostaje jeszcze przypadek, z którym ta wersja sobie nie radzi, gdy pierwszy
argument jest od razu ujemny. Radzi sobie z tym wersja powyżej po prawej.

Okazuje sie֒ jednak, że istnieje proste rozwia֒zanie niewykorzystuja֒ce odcie֒cia,
które rozwia֒zuje wszystkie powyższe problemy:

sum_to( 1, 1 ).

sum_to( N, R ) :-

N > 1,

N1 is N - 1,

sum_to( N1, R1 ),

R is R1 + N.

Odcie֒cie — przypadek 2: stwierdza fa lszywość celu

Odcie֒cie — przypadek 3: odcina niepotrzebne możliwości

Odcie֒cie — problemy

Prolog — operacje na bazie danych

Normalnie w trakcie pracy Prolog jest w trybie odpowiadania na pytania. Fakty
sa֒ dodawane do bazy danych Prologu przez użycie predykatu consult.
(Wczytanie faktów z terminala jest możliwe wywo laniem consult(’user’)).
Dodatkowo, w trakcie pracy można tworzyć nowe fakty dynamicznie, i zarówno
dodawać je do bazy danych, jak również usuwać istnieja֒ce. Powoduje to jakby
samo-modyfikacje֒ programu.

asserta(term), assertz(term) — dodaje fakt term do bazy danych,
odpowiednio na pocza֒tek i na koniec

retract(term) — kasuje fakt term z bazy danych, o ile w niej by l

Uwaga: przy nawracaniu Prologu efekty dzia lania tych operacji nie sa֒ kasowane,
to znaczy poprzednie stany bazy danych nie sa֒ odtwarzane!

Prolog — operacje na termach

var(term) /* prawdziwy jesli term jest niepodstawiona zmienna */

nonvar(term) /* odwrotnie niz var */

atom(term) /* czy term jest podstawionym literalem atomowym (nie stringiem)

integer(term)

atomic(term) /* atom lub integer */

clause /* C&M(4)p.115 */

functor /* C&M(2)p.120(4)p.117 */

arg /* C&M(2)p.122(4)p.119 */

=.. /* C&M(2)p.173,123(4)p.120 */

Prolog — operacje wej́scia/wyj́scia

Wczytywanie i wypisywanie termów:

?- read(X). /* wczytuje z terminala jeden term zakonczony

kropka ’.’ i podstawia go pod zmienna X;

na koncu pliku read zwraca end_of_file */

?- write(X). /* wypisuje na terminalu wartosc termu

aktualnie podstawiona pod zmienna X */

?- nl. /* wypisuje znak nowej linii na terminalu */

Wczytywanie i wypisywanie znaków:

?- get(X) /* czyta znak (w postaci numerycznego kodu znaku */

?- put(X) /* pisze jeden znak, np. put(104) wypisuje znak ’h’ */

Operacje na plikach:

?- tell(’nowy’)./* otwiera nowy plik o nazwie ’nowy’ i przelacza

standardowe wyjscie na ten plik;

nastepne operacje wyjscia dzialaja na tym pliku */

?- told. /* zamyka plik aktualnie otwarty i

przelacza standardowe wyjscie na terminal */

?- see(’stary’). /* otwiera istniejacy plik do odczytu */

?- seen. /* konczy czytanie z pliku i zamyka plik */

Czytanie ca lych plików w trybie definiowania aksjomatów:

?- consult(plik1). /* w skrocie mozna: [plik1] */

?- reconsult(plik2). /* w skrocie mozna: [-plik2] */

Przyk lad: zapisanie na pliku wszystkich aksjomatów definiuja֒cych predykaty
’ma’ i ’lubi’:

?- tell(’program’), listing(ma), listing(lubi), told.

/* predykat listing wypisuje na terminalu wszystkie posiadane

klauzule; np.: listing, listing(ma), listing(ma/2) */

